* **************************************************************************** *
* Sierra Leone - Social Signaling and Childhood Immunization
* Reputational Costs and Benefits in terms of Actions
* **************************************************************************** *
/*
** Purpose:     Reputational Costs and Benefits in terms of Actions

** Written by:
                Anne Karing    (akaring@uchiago.edu)
                Jonas Guthoff  (jonas.guthoff@gmail.com)

*/
* **************************************************************************** *
* **************************************************************************** *


	* Set format/style for figures:
	set scheme plotplainblind
	graph drop _all

	* ----------------------------------------------------------------------------
	* Figure: Motives for Social Signaling

	use  "${Replicate_SocialSignals_dtaInter}/Endline_Survey_Data.dta", clear


	summ  scold instruct fine beat other_miss nothingdo
	local Observations1 = trim("`: display %-9.0gc r(N)'")

	graph bar  ///
	scold instruct fine beat other_miss nothingdo,            ///
	subtitle("missed to take your child for vaccination?")    ///
	ytitle("Share of Respondents")                            ///
	yscale(r(0 1)) ylabel(0(0.2)1) blabel(bar, format(%12.2f))  nopercent ///
	legend(off) ///
	text(-0.03 11 "Scold", size(vsmall))                      ///
	text(-0.03 26.5 "Encourage", size(vsmall))                   ///
	text(-0.03 42 "Fine", size(vsmall))                        ///
	text(-0.03 58 "Beat", size(vsmall))                        ///
	text(-0.03 73 "Other", size(vsmall))                       ///
	text(-0.03 89 "Take no action", size(vsmall))               ///
	name(actionmiss) scheme(plotplainblind) ytick(0.1(0.1)0.9, grid)


	summ  praise gift other_all nothingdo_all
	local Observations2 = trim("`: display %-9.0gc r(N)'")

	graph bar  ////
	praise gift other_all nothingdo_all,                 ///
	subtitle("took your child for all immunizations?")   ///
	ytitle("Share of Respondents")                       ///
	yscale(r(0 1)) ylabel(0(0.2)1) blabel(bar, format(%12.2f))  nopercent ///
	legend(off) ///
	text(-0.03 16  "Praise/appreciate/" "talk good", size(vsmall))                ///
	text(-0.03 38.5 "Gift/" "compensate", size(vsmall))                              ///
	text(-0.03 61 "Other", size(vsmall))                                          ///
	text(-0.03 84 "Take no action", size(vsmall))                                ///
	name(actionall) scheme(plotplainblind) ytick(0.1(0.1)0.9, grid)

	* Combine the graphs:
	graph combine ///
	actionall actionmiss, ///
	subtitle("What action would they take if you...")
	graph export "${Replicate_SocialSignals_AFigures}/Figure_MotivesSocialSignaling_NEW.jpg", replace




* **************************************************************************** *
* End of Figure: Motives for Social Signaling !!!
* **************************************************************************** *
